
D01 – Quadrature

D01BDF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

D01BDF calculates an approximation to the integral of a function over a finite interval [a, b]:

I =
∫ b

a

f(x) dx.

It is non-adaptive and as such is recommended for the integration of ‘smooth’ functions. These exclude
integrands with singularities, derivative singularities or high peaks on [a, b], or which oscillate too strongly
on [a, b].

2 Specification

SUBROUTINE D01BDF(F, A, B, EPSABS, EPSREL, RESULT, ABSERR)
real F, A, B, EPSABS, EPSREL, RESULT, ABSERR
EXTERNAL F

3 Description

D01BDF is based on the QUADPACK routine QNG (Piessens et al. [2]). It is a non-adaptive routine
which uses as its basic rules, the Gauss 10-point and 21-point formulae. If the accuracy criterion is not
met, formulae using 43 and 87 points are used successively, stopping whenever the accuracy criterion is
satisfied.

This routine is designed for smooth integrands only.

4 References

[1] Patterson T N L (1968) The Optimum addition of points to quadrature formulae Math. Comput.
22 847–856

[2] Piessens R, de Doncker–Kapenga E, Überhuber C and Kahaner D (1983) QUADPACK, A
Subroutine Package for Automatic Integration Springer-Verlag

5 Parameters

1: F — real FUNCTION, supplied by the user. External Procedure
F must return the value of the integrand f at a given point.
Its specification is:

real FUNCTION F(X)
real X

1: X — real Input
On entry: the point at which the integrand f must be evaluated.

F must be declared as EXTERNAL in the (sub)program from which D01BDF is called. Parameters
denoted as Input must not be changed by this procedure.

2: A — real Input
On entry: the lower limit of integration, a.

3: B — real Input

On entry: the upper limit of integration, b. It is not necessary that a < b.

4: EPSABS — real Input

On entry: the absolute accuracy required. If EPSABS is negative, the absolute value is used. See
Section 7.

5: EPSREL — real Input

On entry: the relative accuracy required. If EPSREL is negative, the absolute value is used. See
Section 7.

6: RESULT — real Output

On exit: the approximation to the integral I.

7: ABSERR — real Output

On exit: an estimate of the modulus of the absolute error, which should be an upper bound for
|I − RESULT|.

6 Error Indicators and Warnings

There are no specific errors detected by the routine. However, if ABSERR is greater than

max{EPSABS,EPSREL× |RESULT|}

this indicates that the routine has probably failed to achieve the requested accuracy within 87 function
evaluations.

7 Accuracy

The routine attempts to compute an approximation, RESULT, such that:

|I − RESULT| ≤ tol.

where
tol = max{|EPSABS|, |EPSREL| × |I|}

and EPSABS and EPSREL are user-specified absolute and relative error tolerances. There can be no
guarantee that this is achieved, and users are advised to subdivide the interval if they have any doubts
about the accuracy obtained. Note that ABSERR contains an estimated bound on |I − RESULT|.

8 Further Comments

The time taken by the routine depends on the integrand and on the accuracy required.

9 Example

To compute ∫ 1

0

x2 sin(10πx) dx.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* D01BDF Example Program Text
* Mark 14 Revised. NAG Copyright 1989.
* .. Parameters ..

INTEGER NOUT
PARAMETER (NOUT=6)

* .. Scalars in Common ..
real PI
INTEGER KOUNT

* .. Local Scalars ..
real A, ABSERR, B, EPSABS, EPSREL, RESULT

* .. External Functions ..
real FST, X01AAF
EXTERNAL FST, X01AAF

* .. External Subroutines ..
EXTERNAL D01BDF

* .. Intrinsic Functions ..
INTRINSIC ABS, MAX

* .. Common blocks ..
COMMON /TELNUM/PI, KOUNT

* .. Executable Statements ..
WRITE (NOUT,*) ’D01BDF Example Program Results’
PI = X01AAF(0.0e0)
EPSABS = 0.0e0
EPSREL = 1.0e-04
A = 0.0e0
B = 1.0e0
KOUNT = 0

*
CALL D01BDF(FST,A,B,EPSABS,EPSREL,RESULT,ABSERR)

*
WRITE (NOUT,*)
WRITE (NOUT,99999) ’A - lower limit of integration = ’, A
WRITE (NOUT,99999) ’B - upper limit of integration = ’, B
WRITE (NOUT,99998) ’EPSABS - absolute accuracy requested = ’,

+ EPSABS
WRITE (NOUT,99998) ’EPSREL - relative accuracy requested = ’,

+ EPSREL
WRITE (NOUT,*)
WRITE (NOUT,99997) ’RESULT - approximation to the integral = ’,

+ RESULT
WRITE (NOUT,99998) ’ABSERR - estimate to the absolute error = ’,

+ ABSERR
WRITE (NOUT,99996) ’KOUNT - number of function evaluations = ’,

+ KOUNT
WRITE (NOUT,*)
IF (KOUNT.GT.87 .OR. ABSERR.GT.MAX(EPSABS,EPSREL*ABS(RESULT)))

+ THEN
WRITE (NOUT,*)

+ ’Warning - requested accuracy may not have been achieved’
END IF
STOP

*
99999 FORMAT (1X,A,F10.4)
99998 FORMAT (1X,A,e9.2)

99997 FORMAT (1X,A,F9.5)
99996 FORMAT (1X,A,I4)

END
*

real FUNCTION FST(X)
* .. Scalar Arguments ..

real X
* .. Scalars in Common ..

real PI
INTEGER KOUNT

* .. Intrinsic Functions ..
INTRINSIC SIN

* .. Common blocks ..
COMMON /TELNUM/PI, KOUNT

* .. Executable Statements ..
KOUNT = KOUNT + 1
FST = (X**2)*SIN(10.0e0*PI*X)
RETURN
END

9.2 Program Data

None.

9.3 Program Results

D01BDF Example Program Results

A - lower limit of integration = 0.0000
B - upper limit of integration = 1.0000
EPSABS - absolute accuracy requested = 0.00E+00
EPSREL - relative accuracy requested = 0.10E-03

RESULT - approximation to the integral = -0.03183
ABSERR - estimate to the absolute error = 0.13E-10
KOUNT - number of function evaluations = 43

